/*/{Protheus.doc} FSYNCT5
E a User Function de busca de e concatena o lancamento padrao, sequencia e CNPJ do Participante para ser usada na busca padrao SYNCT5
@author Elton Teodoro Alves
@since 14/11/2014
@version P11 R5
/*/
#Include 'Protheus.ch'

Static	TIPO_PART	:=	1
Static	COD_LANPAD	:=	2
Static	CLIENTE		:=	3
Static	LOJA_CLI	:=	4
Static	FORNECEDOR	:=	5
Static	LOJA_FOR	:=	6
Static cLPPart		:=	''

User Function fSYNCT5()
	
	Local		aArea		:=	GetArea()
	Local		aParamBox	:=	{}
	Local		aRet		:=	{}
	Local		aAlias		:=	{'SA1','SA2'}
	Local		cAlias		:=	'CT2'
	Local		cCodPart	:=	''
	Local		cLojPart	:=	''
	Private	cCadastro	:=	''
	
	If SX2->(DbSeek(cAlias))
		
		cCadastro	:=	Capital(X2Nome())
		
	End If
	
	aAdd(aParamBox,{3,'Tipo de Participante',1,{'Cliente','Fornecedor'},50,'',.F.})
	
	aAdd(aParamBox,{1,'Lancamento Padrao',Space(GetSx3Cache('CT5_LANPAD','X3_TAMANHO') + GetSx3Cache('CT5_SEQUEN','X3_TAMANHO') + 1),'!@','','CT5SYN','',0,.F.})
	
	aAdd(aParamBox,{1,'Cliente',Space(GetSx3Cache('A1_COD','X3_TAMANHO')),'!@','','SA1','MV_PAR01 == 1',0,.F.})
	aAdd(aParamBox,{1,'Loja',Space(GetSx3Cache('A1_LOJA','X3_TAMANHO')),'!@','','SA1LJ','MV_PAR01 == 1',0,.F.})
	
	aAdd(aParamBox,{1,'Fornecedor',Space(GetSx3Cache('A2_COD','X3_TAMANHO')),'!@','','SA2A','MV_PAR01 == 2',0,.F.})
	aAdd(aParamBox,{1,'Loja',Space(GetSx3Cache('A2_LOJA','X3_TAMANHO')),'!@','','SA22','MV_PAR01 == 2',0,.F.})
	
	If ParamBox(aParamBox,'Origem',@aRet)
		
		DbSelectArea(aAlias[aRet[TIPO_PART]])
		DbSetOrder(1)
		
		If aAlias[aRet[TIPO_PART]] == aAlias[001]
			
			cCodPart	:=	aRet[CLIENTE]
			cLojPart	:=	aRet[LOJA_CLI]
			
		ElseIf aAlias[aRet[TIPO_PART]] == aAlias[002]
			
			cCodPart	:=	aRet[FORNECEDOR]
			cLojPart	:=	aRet[LOJA_FOR]
			
		End If
		
		cLPPart	:=	AllTrim(aRet[COD_LANPAD]) + '/'
		
		If DbSeek(xFilial(aAlias[aRet[TIPO_PART]]) + cCodPart + cLojPart)			
			
			If (aAlias[aRet[TIPO_PART]])->&('A' + SubStr(aAlias[aRet[TIPO_PART]],3,1) + '_XPARTIC')
				
				cLPPart	+=	(aAlias[aRet[TIPO_PART]])->&('A' + SubStr(aAlias[aRet[TIPO_PART]],3,1) + '_CGC')
				
			End If
			
		End If
		
	End If
	
	RestArea(aArea)
	
Return .T.

/*/{Protheus.doc} RSYNFIL
E a User Function de retorno de filiais para ser usada na busca padrao SYNFIL
@author Elton Teodoro Alves
@since 14/11/2014
@version P11 R5
/*/
User Function rSYNCT5()
Return cLPPart
